
LiveSelect

Dynamic (multi)selection field for LiveView.
LiveSelect is a LiveView component that implements a dynamic selection field with a dropdown. The content of the
dropdown is filled by your LiveView as the user types. This allows you to easily create an 
interface for search-like functionalities with type-ahead. LiveSelects features include:
	Single as well as multiple selection
	Options to configure the behaviour, such as minimum number of characters that trigger an update or the maximum number of selectable options
	Default styles for daisyUI and tailwindcss, which are fully customizable and can be completely overridden if needed
	Ability to customize the rendered HTML for dropdown entries and tags using slots.


  
    
  
  Try it in the showcase app 🔬



  
    
  
  Single selection (single mode)


[image: DEMO]

  
    
  
  Multiple selection (tags mode)


[image: DEMO]

  
    
  
  Multiple selection (quick_tags mode)


[image: DEMO]

  
    
  
  Usage Example 🧭


Template:
  <.form for={@form} phx-change="change">
    <.live_select field={@form[:city_search]} /> 
  </.form>
NOTE: If your form is implemented in a LiveComponent, add phx-target={@myself}, like this:
  <.live_select field={@form[:city_search]} phx-target={@myself} />
In the LiveView or LiveComponent that's the target of your form events:
  @impl true
  def handle_event("live_select_change", %{"text" => text, "id" => live_select_id}, socket) do 
    cities = City.search(text)
    # cities = [ 
    # {"New York City", [-74.00597,40.71427]}, 
    # {"New Kingston", [-76.78319,18.00747]}, 
    # ... 
    # ]

    send_update(LiveSelect.Component, id: live_select_id, options: cities)
    
    {:noreply, socket}
  end

  @impl true
  def handle_event(
        "change",
        %{"my_form" => %{"city_search_text_input" => city_name, "city_search" => city_coords}},
        socket
      ) do
    IO.puts("You selected city #{city_name} located at: #{city_coords}")

    {:noreply, socket}
  end  
Refer to the module documentation for the details, and
check out the cheatsheet for some useful tips.

  
    
  
  Installation 📦


To install, add this to your dependencies:
[
    {:live_select, "~> 1.0"}
]

  
    
  
  Javascript hooks 🪝


LiveSelect relies on Javascript hooks to work. You need to add LiveSelect's hooks to your live socket.
LiveSelect distributes its Javascript code (a single file) in the same way as LiveView, by including an
npm package as part of its hex package.
To include LiveSelect's hooks, add this to your app.js file:
import live_select from "live_select"

// if you don't have any other hooks:
let liveSocket = new LiveSocket("/live", Socket, {params: {_csrf_token: csrfToken}, hooks: live_select})

// if you have other hooks:
const hooks = {
    MyHook: {
        // ...
    },
    ...live_select
}
let liveSocket = new LiveSocket("/live", Socket, {params: {_csrf_token: csrfToken}, hooks})

  
    
  
  If you're using Webpack or another NPM-based builder


If you're using an npm-based builder such as Webpack, you will need to add LiveSelect to the list of your dependencies in your package.json (just as you did with LiveView):
{
  "dependencies": {
    "phoenix": "file:../deps/phoenix",
    "phoenix_html": "file:../deps/phoenix_html",
    "phoenix_live_view": "file:../deps/phoenix_live_view",
    "live_select":  "file:../deps/live_select" // <-- add this line, and add an extra "../" if you're in an umbrella app
  }
}
And then run npm install from your assets folder. You will also need to run npm install --force live_select
whenever you update the LiveSelect hex package in order to get the latest JS code.

  
    
  
  Styling 🎨


LiveSelect supports 3 styling modes:
	tailwind: uses standard tailwind utility classes (the default)
	daisyui: uses daisyUI classes.
	none: no styling at all.

The choice of style is controlled by the style option
in live_select/1.
tailwind and daisyui styles come with sensible defaults which can be selectively extended or completely overridden.
Refer to the Styling section for further details.
⚠️ Attention
Please note the different paths for a standalone or umbrella app.


  
    
  
  tailwind v3


If you're using tailwind or daisyui styles, you need to add one of the following lines to the content section in
your tailwind.config.js:
module.exports = {
    content: [
        //...
        '../deps/live_select/lib/live_select/component.*ex', // <-- for a standalone app
        '../../../deps/live_select/lib/live_select/component.*ex' // <-- for an umbrella app
    ]
    //..
}

  
    
  
  tailwind v4


If you are using tailwind v4+ and are not using a tailwind.config.js file you instead need to add the relevant @source directive to your app.css file:
@source "../../deps/live_select/lib/live_select/component.*ex" /* for a standalone app */
@source "../../../../deps/live_select/lib/live_select/component.*ex" /* for an umbrella app */

  
    
  
  Showcase app 🎪


The repository includes a showcase app that you can use to experiment with the different options and parameters
for LiveSelect.
The showcase app is available here.
To start the showcase app locally, simply run:
mix setup
PORT=4001 mix phx.server
from within the cloned repository. The app will be available at http://localhost:4001. The showcase app allows you to
quickly experiment with options and styles, providing an easy way to fine tune your LiveSelect component. The app also
shows the messages and events that your LiveView receives. For each event or message, the app shows the function head
of the callback that your LiveView needs to implement in order to handle the event.

  
    
  
  Contribute 🤝


Contributions are very welcome! However, if you want do add a new feature please discuss it first by creating an issue so we can all agree that it's needed. 
Also, it's important to add a test that covers it. If you don't know how to write the test or need guidance,
I'm happy to help.
Use mix test to run the entire test suite, which is subdivided into 3 main files:
	test/live_select/component_test.exs - everything that can be tested by rendering the component statically
	test/live_select_test.exs - tests for single mode that require a running LiveView 
	test/live_select_tags_test.exs - tests for tags mode that require a running LiveView
	test/live_select_quick_tags_test.exs - tests for quick_tags mode that require a running LiveView

Tests that require a LiveView use the showcase app as the parent LiveView.

  
    
  
  Roadmap 🛣️


	[X] Add package.json to enable import live_select from "live_select"
	[X] Make sure component classes are included by tailwind
	[X] Enable custom styling
	[X] Rename LiveSelect.render to live_select
	[X] Customizable placeholder
	[X] Enable configuration of styles in the showcase app
	[X] Add support for vanilla tailwind styles
	[X] Enable multiple selection mode
	[X] Expose as function component (and drop LV 0.17 support)
	[X] Add cheatsheet
	[X] Additional multiple selection mode 
	[ ] Add section to document testing strategies



  

    
Styling
    

LiveSelect supports 3 styling modes:
	tailwind: uses standard tailwind utility classes (the default)
	daisyui: uses daisyUI classes.
	none: no styling at all.

The choice of style is controlled by the style option in live_select/1.
tailwind and daisyui styles come with sensible defaults which can be extended or overridden via options.
This is what each default style looks like:

  
    
  
  daisyui:


[image: daisyui example](the actual colors may differ depending on the selected daisyui theme)

  
    
  
  tailwind:


[image: tailwind example]These defaults can be selectively overridden or extended using the appropriate options
to live_select/1.
You can control the style of the following elements:
	The outer container of the live_select component
	The text_input field
	The text_input_selected text field when an option has been selected
	The dropdown that contains the selectable options
	The single selectable option(s)
	The currently active_option
	The clear_button to clear the selection (only if allow_clear is set)
	selected_option. This is an option in the dropdown that has already been selected. It's still visible, but can't be selected again
	available_option. This is an option in the dropdown that has not been selected and is available for selection 
	unavailable_option. This is an option in the dropdown that has not been selected but is not available for selection. This happens when there is a specified maximum number of selectable elements and that number has been reached.

Here's a visual representation of the elements:
[image: styled elements]
In tags and quick_tags mode there are 3 additional stylable elements:
	tag showing the selected options
	tags_container that contains the tags
	clear_tag_button button to remove the tags

[image: styled elements_tags]
For each of these elements there is an {element}_class and for some also an {element}_extra_class option, which can
be used
to override or extend the default CSS classes for the element, respectively. These options accept both strings and lists of strings.
You can't use both options together:
use {element}_class
to completely override the default classes, or use {element}_extra_class to extend the default.
The following table shows the default styles for each element and the options you can use to adjust its CSS classes.
	Element	Default daisyui classes	Default tailwind classes	Class override option	Class extend option
	active_option	active menu-active	bg-gray-600 text-white	active_option_class	
	available_option	cursor-pointer	cursor-pointer hover:bg-gray-400 rounded	available_option_class	
	clear_button	cursor-pointer hidden	cursor-pointer hidden	clear_button_class	clear_button_extra_class
	clear_tag_button	cursor-pointer	cursor-pointer	clear_tag_button_class	clear_tag_button_extra_class
	container	dropdown dropdown-open	h-full relative text-black	container_class	container_extra_class
	dropdown	bg-base-200 dropdown-content menu menu-compact p-1 rounded-box shadow w-full z-[1]	absolute bg-gray-100 inset-x-0 rounded-md shadow top-full z-50	dropdown_class	dropdown_extra_class
	option		px-4 py-1 rounded	option_class	option_extra_class
	selected_option	cursor-pointer font-bold	cursor-pointer font-bold hover:bg-gray-400 rounded	selected_option_class	
	tag	badge badge-primary p-1.5 text-sm	bg-blue-400 flex p-1 rounded-lg text-sm	tag_class	tag_extra_class
	tags_container	flex flex-wrap gap-1 p-1	flex flex-wrap gap-1 p-1	tags_container_class	tags_container_extra_class
	text_input	input input-bordered pr-6 w-full	disabled:bg-gray-100 disabled:placeholder:text-gray-400 disabled:text-gray-400 pr-6 rounded-md w-full	text_input_class	text_input_extra_class
	text_input_selected	input-primary	border-gray-600 text-gray-600	text_input_selected_class	
	unavailable_option	disabled	text-gray-400	unavailable_option_class	

For example, if you want to remove rounded borders from the options, have the active option use white text on a red background,
and use green as a background color for tags instead of blue, render live_select/1
like this:
<.live_select
   form={my_form}
   field={my_field}
   id="live_select"
   mode={:tags}
   placeholder="Search for a city"
   active_option_class="text-white bg-red-800"
   option_extra_class="!rounded"
   tag_extra_class="!bg-blue-400 bg-green-200" />
Selectively removing classes from defaults
You can remove classes included with the style's defaults by using the !class_name notation
in an {element}_extra_class option. For example, if a default style is rounded-lg px-4,
using an extra class option of !rounded-lg text-black will result in the following final class 
being applied to the element:
 px-4 text-black


  

    
Cheatsheet
    


  
    
  
  Style LiveSelect like a Phoenix Core Component, with label and errors


1. Add this to core_components.ex:
def live_select(%{field: %Phoenix.HTML.FormField{} = field} = assigns) do
  assigns =
    assigns
    |> assign(:errors, Enum.map(field.errors, &translate_error(&1)))
    |> assign(:live_select_opts, assigns_to_attributes(assigns, [:errors, :label]))
  
  ~H"""
  <div phx-feedback-for={@field.name}>
    <.label for={@field.id}><%= @label %></.label>
    <LiveSelect.live_select
      field={@field}
      text_input_class={[
        "mt-2 block w-full rounded-lg border-zinc-300 py-[7px] px-[11px]",
        "text-zinc-900 focus:outline-none focus:ring-4 sm:text-sm sm:leading-6",
        "phx-no-feedback:border-zinc-300 phx-no-feedback:focus:border-zinc-400 phx-no-feedback:focus:ring-zinc-800/5",
        "border-zinc-300 focus:border-zinc-400 focus:ring-zinc-800/5",
        @errors != [] && "border-rose-400 focus:border-rose-400 focus:ring-rose-400/10"
      ]}
      {@live_select_opts}
    />
  
    <.error :for={msg <- @errors}><%= msg %></.error>
  </div>
  """
end
2. Then call it this way:
<.live_select field={@form[:city]} label="City" phx-target={@myself} />
You can also pass any of the other LiveSelect options.

  
    
  
  Implementing a simple search functionality



  
    
  
  1. With no options displayed if the user doesn't enter any text


Heex template:
<.live_select field={@form[:locations]} update_min_len={1} phx-focus="clear" />
Live view:
@impl true
def handle_event("live_select_change", %{"id" => id, "text" => text}, socket) do
 options =
   retrieve_locations()
   |> Enum.filter(&(String.downcase(&1) |> String.contains?(String.downcase(text))))

 send_update(LiveSelect.Component, options: options, id: id)

 {:noreply, socket}
end

@impl true
def handle_event("clear", %{"id" => id}, socket) do
  send_update(LiveSelect.Component, options: [], id: id)

  {:noreply, socket}
end

  
    
  
  2. With a fixed set of default options to be displayed when the text input is empty


Heex template:
<.live_select field={@form[:locations]} update_min_len={0} phx-focus="set-default" options={@default_locations} />
Live view:
@impl true
def mount(socket) do
  socket = assign(socket, default_locations: default_locations())
  
  {:ok, socket}
end

@impl true
def handle_event("live_select_change", %{"id" => id, "text" => text}, socket) do
  options =
    if text == "" do 
      socket.assigns.default_locations
    else
      retrieve_locations()
      |> Enum.filter(&(String.downcase(&1) |> String.contains?(String.downcase(text))))
    end
    
  send_update(LiveSelect.Component, options: options, id: id)
    
  {:noreply, socket}
end

@impl true
def handle_event("set-default", %{"id" => id}, socket) do
  send_update(LiveSelect.Component, options: socket.assigns.default_locations, id: id)

  {:noreply, socket}
end

  
    
  
  Dropdown that opens above the input field



  
    
  
  Tailwind


Heex template
<.live_select
   field={@form[:my_field]}
   dropdown_extra_class="!top-full bottom-full" />

  
    
  
  DaisyUI


Heex template
<.live_select
   field={@form[:my_field]}
   style={:daisyui}
   container_extra_class="dropdown-top" />

  
    
  
  Display tags underneath the input field (Tailwind)


Heex template:
<.live_select
   field={@form[:my_field]}
   mode={:tags}
   container_extra_class="flex flex-col"
   dropdown_extra_class="top-11"
   tags_container_extra_class="order-last" />

  
    
  
  Limit the height of the dropdown and make it scrollable (Tailwind)


Heex template:
<.live_select
   field={@form[:my_field]}
   dropdown_extra_class="max-h-60 overflow-y-scroll" />


  

    
CHANGELOG
    


  
    
  
  1.7.4 (2025-11-28)


	Bugfix: quick_tags with overflow does not preserve scroll
	Bugfix: Getting undefined in search box when a tag is unselected


  
    
  
  1.7.3 (2025-10-19)


	Bugfix: label not found in nil


  
    
  
  1.7.2 (2025-10-17)


	Bugfix: label not found in nil
	Bugfix: 2 bugs hit on form recovery


  
    
  
  1.7.1 (2025-09-17)


	Do not show <div> container for tags when not in tagged mode
	Unify blur experience between single and tagged modes


  
    
  
  1.7.0 (2025-06-19)


	DaisyUI 5 compatibility - active options now include both active and menu-active classes for compatibility with DaisyUI 3, 4, and 5
	Add keep_options_on_select flag. By default, when not in quick_tags mode, input text field and list of selectable options are now cleared after selection. This flag suppresses this behavior   
	Positioning and sizing clear buttons entirely using JS
	no_basic_styles_for_clear_buttons flag to suppress basic positioning and styling of clear buttons


  
    
  
  1.6.0 (2025-04-13)


	add ability to disable options


  
    
  
  1.5.5 (2025-03-31)


	make clear buttons also honor the disabled attribute
	remember entered text when blurring away from the element


  
    
  
  1.5.4 (2025-01-29)


	Fix bug causing selection recovery to fail if Phoenix uses the built-in JSON module from Elixir 1.18


  
    
  
  1.5.3 (2025-01-27)


	Fix bug that causes selection recovery to fail when input_for is used to render the component
	Remove dependency from Jason library


  
    
  
  1.5.2 (2024-12-28)


	Fix bug where the keyboard doesn't show on some mobile browsers if an element is already selected in single mode


  
    
  
  1.5.1 (2024-12-28)


	Add unavailable_option_class to style options that cannot be selected because of max_selectable > 0
	Fix can't remove items in quick_tags mode via dropdown with max_selectable


  
    
  
  1.5.0 (2024-12-27)


	new quick_tags mode 
	showcase app: add toggle to style options as checkboxes


  
    
  
  1.4.4 (2024-12-07)


	support for LiveView 1.0.0


  
    
  
  1.4.3 (2024-10-28)


	add options clear_tag_button_class and clear_tag_button_extra_class to style button to clear tags
	fix bug where selection was not restored after focus+blur when value is pre-selected or forced 


  
    
  
  1.4.2 (2024-06-19)


do not restore selection after blur it it was cleard by hitting the clear button

  
    
  
  1.4.1 (2024-06-18)


	fix bug introduced by LV-1.0's new focus behavior (https://github.com/maxmarcon/live_select/issues/72)


  
    
  
  1.4.0 (2024-03-18)


	support for associations and embeds
	add value_mapper assign and decode/1 function


  
    
  
  1.3.3 (2024-02-06)


	add slot to render custom clear button
	add option to extend clear button style


  
    
  
  1.3.2 (2024-01-26)


	updated dependencies, including updating phoenix_html to 4.0.0


  
    
  
  1.3.1 (2023-12-06)


	bugfix: only set selection in client if event was sent to this component


  
    
  
  1.3.0 (2023-12-05)


	added support for selection recovery. Upon reconnection, the client sends an event (selection_recovery) that contains the latest selection. This allows recovery of the selection that was active before the view disconnected.


  
    
  
  1.2.2 (2023-10-21)


	daisyui3-compatible
	arrowUp when there is no active option navigates to the last option
	scroll options into view when they become active


  
    
  
  1.2.1 (2023-10-17)


	fix bug that was causing dropdown to overflow container (https://github.com/maxmarcon/live_select/issues/43)


  
    
  
  1.2.0 (2023-09-25)


	add clear_button_class option to style clear buttons for tags and selection
	various bugfixes and improvements


  
    
  
  1.1.1 (2023-07-21)


	accept sticky flag in an option to prevent it from being removed (https://github.com/maxmarcon/live_select/pull/33)
	when selection becomes empty, an update is triggered with a hidden field named after live_select's field's name

(thanks to https://github.com/shamanime for both changes)

  
    
  
  1.1.0 (2023-06-26)


	add phx-focus and phx-blur options to specify events to be sent to the parent upon focus and blur of the text input field
	send live_select_change event directly from JS hook to save a round-trip
	expects a single field assign of type Phoenix.HTML.FormField instead of separate form and field assigns (which is still supported but soft-deprecated with a warning)


  
    
  
  1.0.4 (2023-05-30)


	Do not use name attribute on non-input elements to prevent LV from crashing
	Change default for update_min_len to 1


  
    
  
  1.0.3 (2023-03-31)


	Programmatically override selection with value assign
	Only clear options if entered text is shorter than update_min_len and user types backspace

Bugfix: fix selection via mouseclick not working when rendering nested elements in the :option slot

  
    
  
  1.0.2 (2023-03-20)


styling options now also accept lists of strings

  
    
  
  1.0.1 (2023-02-18)


Bugfix: fix error when using atom form

  
    
  
  1.0.0 (2023-02-15)


This version introduces the following breaking changes and new features:
	Rendering using a function component <.live_select /> instead of the old function style (<%= live_select ... %>)
	Dropping the message-based update cycle (which used handle_info/2) in favour of an event-based update cycle (which uses handle_event/3). This makes it much easier
and more intuitive to use LiveSelect from another LiveComponent.
	Ability to customize the default rendering of dropdown entries and tags using the :option and :tag slots 

 How to upgrade from version 0.x.x: 
	Instead of rendering LiveSelect in this way: <%= live_select form, field, mode: :tags %>, render it in this way: <.live_select form={form} field={field} mode={:tags} />
	Don't forget to add phx-target={@myself} if you're using LiveSelect from another LiveComponent
	Turn your handle_info/2 implementations into handle_event/3:

Turn this:
def handle_info(%ChangeMsg{} = change_msg, socket) do
     options = retrieve_options(change_msg.text)
    
     send_update(LiveSelect.Component, id: change_msg.id, options: options)
    
     {:noreply, socket}
end
into:
def handle_event("live_select_change", %{"text" => text, "id" => live_select_id}, socket) do
    options = retrieve_options(text)
    
    send_update(LiveSelect.Component, id: live_select_id, options: options)
    
    {:noreply, socket}
end
	If you're rendering LiveSelect in a LiveComponent, you can now place your handle_event/3 callback in the LiveComponent, there's no need to put the update logic in the view anymore


  
    
  
  0.4.1 (2023-02-07)


Bugfix: component now works event when strict Content Security Policy are set

  
    
  
  0.4.0 (2023-01-30)


	add available_option_class configuration option to style options that have not been selected yet
	add user_defined_options configuration option to allow user to enter any tag
	enable assigning a custom id to the component
	enable programmatically clearing of the selection via clear assign
	add allow_clear configuration option. If set, an x button will appear next to the text input in single mode. Clicking the button clears the selection


  
    
  
  Deprecations


LiveSelect.update_options/2 has been deprecated in favor of directly using Phoenix.LiveView.send_update/3 

  
    
  
  0.3.3 (2023-01-15)


	set initial selection from the form or manually with value option
	set initial list of available options with options option
	add a max_selectable option to limit the maximum size of the selection


  
    
  
  0.3.2 (2023-01-03)


Bugfix: options in dropdown not always clickable because of race condition with blur event (https://github.com/maxmarcon/live_select/issues/7)

  
    
  
  0.3.1 (2022-12-15)


Bugfix: removed inputs_for because it was failing if the field is not an association

  
    
  
  0.3.0 (2022-12-15)


	tags mode
	hide dropdown on escape key pressed


  
    
  
  0.2.1 (2022-10-25)


	when disabled option is used, also disable hidden input
	style disabled text input in tailwind mode
	fix problem with selection via mouseclick when an input field is underneath the dropdown
	hide dropdown when user clicks away from component or input loses focus
	show dropdown when input obtains focus again
	using default black text in tailwind mode 


  
    
  
  0.2.0 (2022-10-03)


	support for tailwind styles (now the default)
	more opinionated default styles
	ability to selectively remove classes from style defaults using the !class_name notation
	rename option search_term_min_length to update_min_len
	better error messages
	various improvements to the showcase app


  
    
  
  0.1.4 (2022-09-20)


	raise if class and extra_class options are used in invalid combinations (https://github.com/maxmarcon/live_select/issues/2)


  
    
  
  Bugfixes


	route server events to the right live select component using the component id (https://github.com/maxmarcon/live_select/issues/1)


  
    
  
  0.1.3 (2022-08-12)


	rename LiveSelect.update/2 to LiveSelect.update_options/2
	add debounce option
	add search delay option to showcase app
	JSON-encode option values before assigning them to the hidden input field
	add LiveSelect.ChangeMsg struct to be used as change message


  
    
  
  0.1.2 (2022-08-10)


	Disable input field via options
	Placeholder text via options
	Improve docs and showcase app
	Remove setting component id via options


  
    
  
  Bugfixes


	Use atoms as field names, because strings are not accepted by Ecto forms


  
    
  
  0.1.1 (2022-08-09)


	Remove all colors from default daisyui styles
	Improve styling of showcase app
	Improve docs
	Remove the msg_prefix option in favor of change_msg


  
    
  
  0.1.0 (2022-08-09)


First version 🎉


  

    
LiveSelect 
    



      
The LiveSelect component is rendered by calling the live_select/1 function and passing it a form field.
LiveSelect creates a text input field in which the user can type text, and hidden input field(s) that will contain the value of the selected option(s).
Whenever the user types something in the text input, LiveSelect triggers a live_select_change event for your LiveView or LiveComponent.
The message has a text parameter containing the current text entered by the user, as well as id and field parameters with the id of the
LiveSelect component and the name of the LiveSelect form field, respectively.
Your job is to handle the event, retrieve the list of selectable options and then call Phoenix.LiveView.send_update/3
to send the list of options to LiveSelect. See the "Examples" section below for details, and check out the
cheatsheet for some useful tips.
Selection can happen either using the keyboard, by navigating the options with the arrow keys and then pressing enter, or by
clicking an option with the mouse.
Whenever an option is selected, LiveSelect will trigger a standard phx-change event in the form. See the "Examples" section
below for details on how to handle the event. In single and tags mode, the content of the input text field and the list of selectable options are cleared on selection.
To suppress this behavior, use the keep_options_on_select flag in the assigns.
In single mode, if the configuration option allow_clear is set, the user can manually clear the selection by clicking on the x button on the input field.
In tags mode, single tags can be removed by clicking on them.

  
    
  
  Single mode


[image: demo]
  
    
  
  Tags mode


[image: demo]When :tags mode is enabled LiveSelect allows the user to select multiple entries. The entries will be visible above the text input field as removable tags.
The selected entries will be passed to your live view's change and submit event handlers as a list of entries, just like an HTML <select> element with multiple attribute would do.

  
    
  
  Quick tags mode


[image: demo]When :quick_tags mode is enabled, the user can select multiple entries, but the dropdown stays open after selection.
This allows the user to select additional entries in quick succession. Also, the entries can be deselected via the dropdown (as well as by clicking on the removable tags).
The dropdown closes when the LiveSelect element loses focus.

  
    
  
  Options


You can set the initial list of options the user can choose from with the options assign.
Afterwards, you can update the options at any time using Phoenix.LiveView.send_update/3.
Each option will be assigned a label, which will be shown in the dropdown, and a value, which will be the value of the
LiveSelect input when the option is selected.
options can be any enumeration of the following elements:
	atoms, strings or numbers: In this case, each element will be both label and value for the option
	tuples: {label, value} corresponding to label and value for the option
	maps: %{label: label, value: value} or %{value: value}
	keywords: [label: label, value: value] or [value: value]

Options can also be disabled when passing in tuples, maps or keywords. Disabled options are displayed, but can't be selected.
Maps and keywords need a :disabled key with a boolean value, and tuples should be a 3 element tuple of {label, value, is_disabled}
In the case of maps and keywords, if only value is specified, it will be used as both value and label for the option.
Because you can pass a list of tuples, you can use maps and keyword lists to pass the list of options, for example:
%{Red: 1, Yellow: 2, Green: 3}
Will result in 3 options with labels :Red, :Yellow, :Green and values 1, 2, and 3.
Note that the option values, if they are not strings, will be JSON-encoded. Your LiveView will receive this JSON-encoded version in the phx-change and phx-submit events.

  
    
  
  Styling


LiveSelect supports 3 styling modes:
	tailwind: uses standard tailwind utility classes (the default)
	daisyui: uses daisyUI classes.
	none: no styling at all.

Please see the styling section for details

  
    
  
  Alternative tag labels


Sometimes, in :tags mode, you might want to use alternative labels for the tags. For example, you might want the labels in the tags to be shorter
in order to save space. You can do this by specifying an additional tag_label key when passing options as map or keywords. For example, passing these options:
[%{label: "New York", value: "NY", tag_label: "NY"}, %{label: "Barcelona", value: "BCN", tag_label: "BCN"}]
will result in "New York" and "Barcelona" being used for the options in the dropdown, while "NY" and "BCN" will be used for the tags (and the values).

  
    
  
  Sticky options


Add  the sticky: true flag to an option to prevent it from being removed from the selection:
[%{label: "New York", value: "NY", sticky: true}]
Now, whenever the selection contains "New York", the option will stick and the user won't be able to remove it.

  
    
  
  Slots



  
    
  
  Options and tags


You can control how your options and tags are rendered by using the :option and :tag slots.
Both slots will be passed an option as argument. In the case of the :option slot, the option will have an
extra boolean field :selected, which will be set to true if the option has been selected by the user.
Let's say you want to show some fancy icons next to each option in the dropdown and the tags:
<.live_select
        field={@form[:city_search]}
        phx-target={@myself}
        mode={:tags}
      >
        <:option :let={option}>
          <div class="flex">
            <.globe />&nbsp;<%= option.label %>
          </div>
        </:option>
        <:tag :let={option}>
            <.check />&nbsp;<%= option.label %>
        </:tag>
</.live_select>
Here's the result:
[image: slots]
  
    
  
  Clear button


You can use the :clear_button slot to directly specify the inner HTML
of the button that clears the tags or the single selection (if allow_clear is set).
Example:
<.live_select
    field={@form[:city_search]}
    phx-target={@myself}
  >
    <:clear_button>
      This my fancy clear button
    </:clear_button>
</.live_select>

  
    
  
  Controlling the selection programmatically


You can always control the selection programmatically, overriding the current user-selected values,
by sending a :value update to LiveSelect via Phoenix.LiveView.send_update/3:
send_update(LiveSelect.Component, id: live_select_id, value: new_selection)
new_selection must be a single element in :single mode, a list in :tags mode. If it's nil, the selection will be cleared.
After updating the selection, LiveSelect will trigger a change event in the form.
To set a custom id for the component to use with Phoenix.LiveView.send_update/3, you can pass the id assign to live_select/1.

  
    
  
  Examples


These examples describe all the moving parts in detail. You can see these examples in action, see which messages and events are being sent, and play around
with the configuration easily with the showcase app.

  
    
  
  Single mode


The user can search for cities.
The LiveSelect main form input is called city_search.
When a city is selected, the coordinates of that city will be the value of the form input.
The name of the selected city is available in the text input field named city_search_text_input.
Template:
<.form for={@form} phx-change="change">
  <.live_select field={@form[:city_search]} />
</.form>
Forms implemented in LiveComponents
If your form is implemented in a LiveComponent and not in a LiveView, you have to add the phx-target attribute
when rendering LiveSelect:
 <.live_select field={@form[:city_search]} phx-target={@myself} />
LiveView or LiveComponent that is the target of the form's events:
@impl true
def handle_event("live_select_change", %{"text" => text, "id" => live_select_id}, socket) do
    cities = City.search(text)
    # cities could be:
    # [ {"city name 1", [lat_1, long_1]}, {"city name 2", [lat_2, long_2]}, ... ]
    #
    # but it could also be (no coordinates in this case):
    # [ "city name 1", "city name 2", ... ]
    #
    # or:
    # [ [label: "city name 1", value: [lat_1, long_1]], [label: "city name 2", value: [lat_2, long_2]], ... ]
    #
    # or even:
    # ["city name 1": [lat_1, long_1], "city name 2": [lat_2, long_2]]

    send_update(LiveSelect.Component, id: live_select_id, options: cities)

    {:noreply, socket}
end

@impl true
def handle_event(
    "change",
    %{"my_form" => %{"city_search_text_input" => city_name, "city_search" => city_coords}},
    socket
  ) do
    IO.puts("You selected city #{city_name} located at: #{city_coords}")

    {:noreply, socket}
end

  
    
  
  Tags mode


Let's say you want to build on the previous example and allow the user to select multiple cities and not only one.
The :tags mode allows you to do exactly this.
Template:
<.form for={@form} phx-change="change">
  <.live_select field={@form[:city_search]} mode={:tags} />
</.form>
LiveView or LiveComponent that is the target of the form's events:
@impl true
def handle_event(
    "change",
    %{"my_form" => %{"city_search" => list_of_coords}},
    socket
  ) do
  # list_of_coords will contain the list of the JSON-encoded coordinates of the selected cities, for example:
  # ["[-46.565,-23.69389]", "[-48.27722,-18.91861]"]

  IO.puts("You selected cities located at: #{list_of_coords}")

  {:noreply, socket}
end

  
    
  
  Multiple LiveSelect inputs in the same LiveView


If you have multiple LiveSelect inputs in the same LiveView, you can distinguish them based on the field id.
For example:
Template:
<.form for={@form} phx-change="change">
  <.live_select field={@form[:city_search]} />
  <.live_select field={@form[:album_search]} />
</.form>
LiveView or LiveComponent:
@impl true
def handle_event("live_select_change", %{"text" => text, "id" => live_select_id, "field" => live_select_field}, socket) do
  options =
    case live_select_field do
      "form_name_city_search" -> City.search(text)
      "form_name_album_search" -> Album.search(text)
    end

  send_update(LiveSelect.Component, id: live_select_id, options: options)

  {:noreply, socket}
end

  
    
  
  Using LiveSelect with associations and embeds


LiveSelect can also be used to display and select associations or embeds without too much effort.
Let's say you have the following schemas:
defmodule City do
  @moduledoc false

  use Ecto.Schema

  import Ecto.Changeset

  @primary_key false
  embedded_schema do
    field(:name)
    field(:pos, {:array, :float})
  end

  def changeset(%__MODULE__{} = schema \\ %__MODULE__{}, params) do
    cast(schema, params, [:name, :pos])
  end
end

defmodule CitySearchForm do
  use Ecto.Schema

  import Ecto.Changeset

  embedded_schema do
    embeds_many(:city_search, City, on_replace: :delete)
  end

  def changeset(schema \\ %__MODULE__{}, params) do
    cast(schema, params, [])
    |> cast_embed(:city_search)
  end
end
Each city has a name and an array with coordinates - we want LiveSelect to display the name as label in the dropdown and in the tags, but we want
the entire data structure (name + coordinates) to be sent to the server when the user selects.
In order for this to work, we need to:
	Map City structs to the options expected by LiveSelect
	Decode City JSON objects sent by the client

We do (1) by passing a value_mapper assign to LiveSelect. This is a 1-arity function that expects the struct and maps it to the option that LiveSelect should use:
<.live_select field={@form[:city_search]} value_mapper={&value_mapper/1} mode={:tags} />
defp value_mapper(%City{name: name} = value) do
  %{label: name, value: value}
end
As you can see, the label is the name of the city whereas the value is the entire struct. This is because we want to be able to recreate the struct from the value, so we need everything. LiveSelect
uses value_mapper/1 to map the values set in the form to the options expected by LiveSelect.
You can also use the value_mapper/1 function to map values to options when updating the list of options while handling live_select_change:
def handle_event("live_select_change", %{"text" => text, "id" => live_select_id}, socket) do
  options =
    retrieve_options()
    |> Enum.map(&value_mapper/1)

  send_update(LiveSelect.Component, id: live_select_id, options: options)

  {:noreply, socket}
end
IMPORTANT: the output of the value_mapper/1 function should be JSON-encodable
Finally, in order to take care of (2) you need to decode the JSON-encoded list of options that's coming from the client before you can
cast them to create a changeset. To do so, LiveSelect offers a convenience function called LiveSelect.decode/1:
def handle_event("change", params, socket) do
  # decode will JSON-decode the value in city_search, handling the type of selection
  # and taking care of special values such as "" and nil
  params = update_in(params, ~w(city_search_form city_search), &LiveSelect.decode/1)

  # now we can cast the params:
  changeset = CitySearchForm.changeset(params)

  {:noreply, assign(socket, form: to_form(changeset))}
end
That's it! Now Your form with embeds selected and displayed with LiveSelect should work

